



A functional correspondence between monadic evaluators and abstract machines for languages with computational effects

Mads Sig Ager, Olivier Danvy*, Jan Midtgaard

*BRICS¹, Department of Computer Science, University of Aarhus,
IT-parken, Aabogade 34, DK-8200 Aarhus N, Denmark*

Abstract

We extend our correspondence between evaluators and abstract machines from the pure setting of the λ -calculus to the impure setting of the computational λ -calculus. We show how to derive new abstract machines from monadic evaluators for the computational λ -calculus. Starting from (1) a generic evaluator parameterized by a monad and (2) a monad specifying a computational effect, we inline the components of the monad in the generic evaluator to obtain an evaluator written in a style that is specific to this computational effect. We then derive the corresponding abstract machine by closure-converting, CPS-transforming, and defunctionalizing this specific evaluator. We illustrate the construction with the identity monad, obtaining the CEK machine, and with a lifted state monad, obtaining a variant of the CEK machine with error and state.

In addition, we characterize the tail-recursive stack inspection presented by Clements and Felleisen as a lifted state monad. This enables us to combine this stack-inspection monad with other monads and to construct abstract machines for languages with properly tail-recursive stack inspection and other computational effects. The construction scales to other monads—including one more properly dedicated to stack inspection than the lifted state monad—and other monadic evaluators.

© 2005 Elsevier B.V. All rights reserved.

* Corresponding author. Tel.: +45 89 42 93 39; fax: +45 89 42 56 01.

E-mail addresses: mads@brics.dk (M.S. Ager), danvy@brics.dk (O. Danvy), jmi@brics.dk (J. Midtgaard).

¹ Basic Research in Computer Science (www.brics.dk), funded by the Danish National Research Foundation.

Keywords: λ -calculus; Interpreters; Abstract machines; Closure conversion; Transformation into continuation-passing style (CPS); Defunctionalization; Monads; Effects; Proper tail recursion; Stack inspection

1. Introduction

Diehl, Hartel, and Sestoft’s overview of abstract machines for programming-language implementation [15] concluded on the need to develop a theory of abstract machines. In previous work [3,10], we have attempted to contribute to this theory by identifying a correspondence between interpreters (i.e., evaluation functions in the sense of denotational semantics) and abstract machines (i.e., transition systems in the sense of operational semantics). The correspondence builds on the observation that *defunctionalized continuation-passing evaluators are abstract machines*. One can therefore obtain an abstract machine, i.e., a transition system [30], by CPS-transforming and defunctionalizing an evaluator. More generally, any recursive function that is defined over an inductive data type can be turned into a transition system by CPS transformation and defunctionalization. Let us first illustrate the correspondence with the factorial function and the corresponding transition system.

1.1. Example: the factorial function

Here is the factorial function, expressed in Standard ML [27]:

```
(* main0 : int -> int *)
fun main0 n
  = fac0 n
(* fac0 : int -> int *)
and fac0 0
  = 1
  | fac0 n
  = n * (fac0 (n - 1))
```

The definition above is in direct style. We transform it into CPS [11,29,35] by naming each intermediate result, sequentializing their computation, and introducing an extra functional argument, the continuation:

```
(* main1 : int -> int *)
fun main1 n
  = fac1 (n, fn a => a)
(* fac1 : int * (int -> int) -> int *)
and fac1 (0, k)
  = k 1
```

```
| fac1 (n, k)
  = fac1 (n - 1, fn v => k (n * v))
```

In this CPS program, as in all CPS programs, all calls are tail calls and all subcomputations are elementary (i.e., they cannot diverge).

Defunctionalizing the continuation amounts to changing its representation and replacing it by a data type and the corresponding apply function [12,33]. We enumerate all the constructors (i.e., lambda-abstractions) that give rise to inhabitants of this function space. There are two such constructors: the initial continuation in `main` and the continuation in the induction case of `fac`. These two constructors are consumed when the continuation is applied, which happens in both clauses of `fac`—one immediately and the other one in the continuation. The data type representing the continuation therefore has two constructors, and the corresponding apply function has two clauses:

```
datatype cont = C0
              | C1 of int * cont

(* apply_cont : cont * int -> int *)
fun apply_cont (C0, v)
  = v
  | apply_cont (C1 (n, k), v)
  = apply_cont (k, n * v)
```

The first constructor is nullary (i.e., constant) and the second is binary, reflecting the number of free variables in the corresponding lambda-abstractions.

In the defunctionalized factorial function, the continuation is constructed using `C0` and `C1`, and consumed using `apply_cont`:

```
(* main2 : int -> int *)
fun main2 n
  = fac2 (n, C0)

(* fac2 : int * cont -> int *)
and fac2 (0, k)
  = apply_cont (k, 1)
  | fac2 (n, k)
  = fac2 (n - 1, C1 (n, k))
```

This program is first order because it is defunctionalized. All of its calls are tail calls and all of its subcomputations are elementary because it is a (defunctionalized) CPS program. Therefore it is a transition system—i.e., an abstract machine—in the sense of Plotkin [30]: for each function, its actual parameters define a configuration and each of its clauses defines a transition.

For clarity, we can reformat this transition system in a more traditional way:

- Input (integer): n
- Output (integer): v
- Defunctionalized continuations: $k ::= C_0 \mid C_1(n, k)$

- Initial transition, transition rules (two kinds), and final transition:

$n \Rightarrow_{init} \langle n, C_0 \rangle$
$\langle 0, k \rangle \Rightarrow_{fac} \langle k, 1 \rangle$
$\langle n, k \rangle \Rightarrow_{fac} \langle n - 1, C_1(n, k) \rangle$
$\langle C_1(n, k), v \rangle \Rightarrow_{cont} \langle k, n \times v \rangle$
$\langle C_0, v \rangle \Rightarrow_{final} v$

1.2. The functional correspondence

This relation between defunctionalized continuation-passing evaluators and abstract machines suggests a functional correspondence between evaluators and abstract machines [3,10]. This correspondence is constructive: to obtain an abstract machine, we start from a compositional evaluator and


- (1) make it operate on first-order data by closure-converting its expressible and denotable values [25,36];
- (2) sequentialize evaluation by CPS-transforming it [11,29,35], thereby materializing its control flow into continuations; and
- (3) make it operate on first-order control by defunctionalizing these continuations [12,33].

The correspondence originates in Reynolds’s seminal article “Definitional Interpreters for Higher-Order Programming Languages” [33], where all the elements (closure conversion, CPS transformation, and defunctionalization) are presented and used. Today, these elements are classical, textbook material [16,21]. Nevertheless, this correspondence has let us derive Krivine’s machine from a canonical call-by-name evaluator and Felleisen et al.’s CEK machine from a canonical call-by-value evaluator. These two machines have been independently developed. To the best of our knowledge, and with the exception of Felleisen and Friedman’s initial presentation of the CEK machine [17, Section 2], these two machines have never been associated with defunctionalization, CPS transformation, and closure conversion. The correspondence has also let us reveal the evaluators underlying Landin’s SECD machine, Schmidt’s VEC machine, Hannan and Miller’s CLS machine, and Curien et al.’s Categorical Abstract Machine [3,10].

We have verified that the correspondence holds for call-by-need evaluators and lazy abstract machines [4], logic programming [7], imperative programming, and object-oriented programming, including Featherweight Java and a subset of Smalltalk. We have also constructed generalizations of Krivine’s machine and of the CEK machine by starting from normalization functions [2]. The correctness of the abstract machines (resp. of the evaluators) is a corollary of the correctness of the evaluators (resp. of the abstract machines) and of the correctness of the transformations.

In this article, we take a next step by applying the methodology to evaluators and abstract machines for languages with computational effects [6,28,37]. We consider a generic evaluator parameterized by a monad (Sections 2 and 3). We then successively consider two monads: the identity monad (Section 4) and a lifted state monad (Section 5). We inline the components of these monads in the generic evaluator, obtaining two specific evaluators. The first one is in direct style, reflecting the computational effect of the identity

monad. The second one is in state-passing style, reflecting the computational effect of the state monad. We then construct the corresponding abstract machines by closure-converting, CPS-transforming, and defunctionalizing these specific evaluators:



We next turn to the security technique of ‘stack inspection’ [20]. Clements and Felleisen recently debunked the myth that stack inspection is incompatible with proper tail recursion [8]. To this end, they presented an abstract machine implementing stack inspection in a properly tail-recursive way. We characterize Clements and Felleisen’s stack inspection as a lifted state monad (Section 6). We then present a monad that accounts for stack inspection more precisely than the lifted state monad, we review related work, and we conclude.

In an extended version of this article [5], we also consider a lifting monad, a state monad, an exception monad, and the two possible monads obtained by combining the state and exception monads. We mechanically construct the corresponding abstract machines. Space, however, prevents us to go into detail here.

2. A call-by-value monadic evaluator in ML

As traditional [6,18,37], we specify a monad as a type constructor and two polymorphic functions:

```

signature MONAD
= sig
  type 'a monad

  val unit : 'a -> 'a monad
  val bind : 'a monad * ('a -> 'b monad) -> 'b monad
end
  
```

Our source language is the untyped λ -calculus with integer literals:

```

datatype term = LIT of int
              | VAR of ide
  
```

```

      | LAM of ide * term
      | APP of term * term

```

where identifiers are represented as values of type `ide`. Programs are closed terms.

The corresponding expressible values are integers and functions:

```

datatype value = NUM of int
               | FUN of value -> value M.monad

```

for a structure `M : MONAD`.

Our monadic interpreter uses an environment `Env` with the following signature:

```

signature ENV
= sig
  type 'a env

  val empty : 'a env
  val extend : ide * 'a * 'a env -> 'a env
  val lookup : ide * 'a env -> 'a
end

```

Throughout this article e denotes environments and e_{empty} denotes the empty environment.

The evaluation function is defined by structural induction on terms:

```

(* eval : term * value Env.env -> value M.monad *)
fun eval (LIT i, e)
  = M.unit (NUM i)
| eval (VAR x, e)
  = M.unit (Env.lookup (x, e))
| eval (LAM (x, t), e)
  = M.unit (FUN (fn v => eval (t, Env.extend (x, v, e))))
| eval (APP (t0, t1), e)
  = M.bind (eval (t0, e),
            fn v0 => M.bind (eval (t1, e),
                            fn v1 => let val (FUN f) = v0
                                    in f v1
                                    end))

```

Given a program, the main evaluation function calls `eval` with this term and the initial environment:

```

fun main t
  = eval (t, env_base)

```

In actuality, this evaluation function, `eval`, `env_base`, and `value` are defined in an ML functor parameterized with a structure `M : MONAD`.

Except for the identity monad, each monad comes with operations that need to be integrated in the source language. Rather than systematically extending the syntax of the

source language with these operations, we hold some of them in the initial environment. For example, rather than having a special form for the successor function, we define it with a binding in the base environment:

```
val env_base
    = Env.extend ("succ", FUN (fn (NUM i)
                               => M.unit (NUM (i + 1))), Env.empty)
```

3. On using ML as a metalanguage

ML is a Turing-complete, statically typed, call-by-value language with computational effects:

- ML programs can therefore diverge and to this end, ML comes with a ‘built-in’ lifting monad to account for divergence. In Section 2, we implicitly make use of this monad in the codomain of `eval`: applying `eval` to a term and an environment only yields a result if it terminates.
- Compiling the evaluator of Section 2 yields warnings to the effect that pattern matching, in the clause for `APP` and in the initial environment, is incomplete. Should we attempt to evaluate a source program that is ill-typed (e.g., because it applies the successor function to a function instead of to an integer), a run-time exception would be raised. In that sense, ML also comes with a ‘built-in’ error monad to account for pattern-matching errors.

In the remainder of this article, we instantiate the evaluator of Section 2 with monads. We could be pedantic and only consider monads that are layered on top of two lifting monads—one for pattern-matching errors and one for divergence. The result would be a notational overkill, and therefore we choose to use ML’s built-in monads.

For the purpose of our work, we view monads as a factorization device for writing evaluators, as in Wadler’s tutorial [37]. We symbolically simplify the monadic evaluator of Section 2 with respect to a given monad (thereby obtaining a direct-style evaluator out of the identity monad, a lifted evaluator out of the lifting monad, a state-threading evaluator out of a state monad, a continuation-passing evaluator out of the continuation monad, an exception-oriented evaluator out of an exception monad, etc.). Our symbolic simplification undoes Moggi’s factorization and it is carried out by inlining the definitions of the type constructor, of `unit` and `bind`, and of the monadic operations.

Finally, we follow the functional-programming tradition initiated by Wadler [37] and we reason equationally over the definitions of `unit` and `bind` to verify that they satisfy the three monadic laws:

- Left unit: `bind (unit a, k) = k a`
- Right unit: `bind (m, unit) = m`
- Associativity: `bind (m, fn a => bind (k a, h)) = bind (bind (m, k), h)`

4. From the identity monad to an abstract machine

We first specify the identity monad and inline its components in the monadic evaluator of Section 2, obtaining an evaluator in direct style. We then take the same steps as in our

previous work [3]: closure conversion, CPS transformation, and defunctionalization. The result is Felleisen et al.'s CEK machine [17,19].

4.1. The identity monad

The identity monad is specified with an identity type constructor and the corresponding two polymorphic functions:

```
structure Identity_Monad : MONAD
= struct
  type 'a monad = 'a

  fun unit a = a
  fun bind (m, k) = k m
end
```

4.2. Inlining the monad in the monadic evaluator

Inlining the components of the identity monad in the monadic evaluator of Section 2 yields an ordinary call-by-value evaluator in direct style where numerals are mapped to numbers, variables are mapped to their denotation, etc.:

```
datatype value = NUM of int
              | FUN of value -> value

val env_base
  = Env.extend ("succ", FUN (fn (NUM i)=>(NUM (i+1))), Env.empty)

(* eval : term * value Env.env -> value *)
fun eval (LIT i, e)
  = NUM i
| eval (VAR x, e)
  = Env.lookup (x, e)
| eval (LAM (x, t), e)
  = FUN (fn v => eval (t, Env.extend (x, v, e)))
| eval (APP (t0, t1), e)
  = let val v0 = eval (t0, e)
        val v1 = eval (t1, e)
        val (FUN f) = v0
      in f v1
    end

fun main p
  = eval (p, env_base)
```

4.3. Closure conversion

We defunctionalize the function space in the data type of values. There are two function constructors:

- one in the denotation of lambda-abstractions, which we represent by a closure, pairing the code of lambda-abstractions together with their lexical environment, and
- one in the initial environment, which we represent by a specialized constructor SUCC.

We splice these two constructors in the data type of values:

```
datatype value = NUM of int
              | CLO of ide * term * value Env.env
              | SUCC
```

Closures are produced when interpreting lambda-abstractions, and the successor function is produced in the initial environment. Both are consumed when interpreting applications. The rest of the evaluator does not change:

```
val env_base = Env.extend ("succ", SUCC, Env.empty)

(* eval : term * value Env.env -> value *)
fun eval (LIT i, e)
  = NUM i
  | eval (VAR x, e)
  = Env.lookup (x, e)
  | eval (LAM (x, t), e)
  = CLO (x, t, e)
  | eval (APP (t0, t1), e)
  = let val v0 = eval (t0, e)
      val v1 = eval (t1, e)
      in case v0
        of (CLO (x, t, e))
         => eval (t, Env.extend (x, v1, e))
        | SUCC
         => let val (NUM i) = v1
            in NUM (i + 1)
            end
      end
  end
fun main p
  = eval (p, env_base)
```

4.4. CPS transformation

We materialize the control flow of the evaluator using continuations. The data type of values and the initial environment do not change. The evaluation function takes an extra parameter, the continuation. Values that used to be returned in the direct-style evaluator are now passed to the continuation. Intermediate values that used to be named with a let

expression are now named by the parameter of a new continuation:

```
(* eval : term * value Env.env * (value -> 'a) -> 'a *)
fun eval (LIT i, e, k)
  = k (NUM i)
  | eval (VAR x, e, k)
  = k (Env.lookup (x, e))
  | eval (LAM (x, t), e, k)
  = k (CLO (x, t, e))
  | eval (APP (t0, t1), e, k)
  = eval (t0, e, fn v0 =>
    eval (t1, e, fn v1 =>
      (case v0
        of (CLO (x, t, e))
         => eval (t, Env.extend (x, v1, e), k)
        | SUCC
         => let val (NUM i) = v1
              in k (NUM (i + 1))
            end)))

fun main p
  = eval (p, env_base, fn v => v)
```

The same evaluator is obtained by inlining the components of the continuation monad in the monadic evaluator of Section 2 and closure-converting the result.

4.5. Defunctionalization

We defunctionalize the function space of continuations. There are three function constructors:

- one in the initial continuation, which we represent by a constructor STOP, and
- two in the interpretation of applications, one with $t1$, e , and k as free variables, and one with $v0$ and k as free variables.

We represent the function space of continuations with a data type with three constructors and an apply function interpreting these constructors. As already noted elsewhere [12,13], the data type of defunctionalized continuations coincides with the data type of evaluation contexts for the source language [16,17]:

```
datatype cont = STOP
  | ARG of term * value Env.env * cont
  | FUN of value * cont
```

The data type of values and the initial environment do not change. Continuations that used to be constructed with a function abstraction in the continuation-passing evaluator are now constructed with STOP, ARG, or FUN. Continuations that used to be consumed with a function application are now consumed by the dispatching function `apply_cont`:

```
(* eval : term * value Env.env * cont -> value *)
fun eval (LIT i, e, k)
```

```

    = apply_cont (k, NUM i)
  | eval (VAR x, e, k)
    = apply_cont (k, Env.lookup (x, e))
  | eval (LAM (x, t), e, k)
    = apply_cont (k, CLO (x, t, e))
  | eval (APP (t0, t1), e, k)
    = eval (t0, e, ARG (t1, e, k))
(* apply_cont : cont * value -> value *)
and apply_cont (STOP, v)
  = v
  | apply_cont (ARG (t1, e, k), v0)
    = eval (t1, e, FUN (v0, k))
  | apply_cont (FUN (CLO (x, t, e), k), v)
    = eval (t, Env.extend (x, v, e), k)
  | apply_cont (FUN (SUCC, k), NUM i)
    = apply_cont (k, NUM (i + 1))

fun main p
  = eval (p, env_base, STOP)

```

This defunctionalized continuation-passing evaluator is an implementation of the CEK machine extended with literals [17,19], which we present next.

4.6. The CEK machine

- Source syntax (terms):

$$t ::= \ulcorner i \urcorner \mid x \mid \lambda x.t \mid t_0 t_1$$

- Expressible values (integers, closures, and predefined functions) and evaluation contexts (i.e., defunctionalized continuations):

$$v ::= i \mid [x, t, e] \mid succ$$

$$k ::= stop \mid fun(v, k) \mid arg(t, e, k)$$

- Initial transition, transition rules (two kinds), and final transition:

$t \Rightarrow_{init} \langle t, e_{init}, stop \rangle$
$\langle \ulcorner i \urcorner, e, k \rangle \Rightarrow_{eval} \langle k, i \rangle$
$\langle x, e, k \rangle \Rightarrow_{eval} \langle k, e(x) \rangle$
$\langle \lambda x.t, e, k \rangle \Rightarrow_{eval} \langle k, [x, t, e] \rangle$
$\langle t_0 t_1, e, k \rangle \Rightarrow_{eval} \langle t_0, e, arg(t_1, e, k) \rangle$
$\langle arg(t_1, e, k), v \rangle \Rightarrow_{cont} \langle t_1, e, fun(v, k) \rangle$
$\langle fun([x, t, e], k), v \rangle \Rightarrow_{cont} \langle t, e[x \mapsto v], k \rangle$
$\langle fun(succ, k), i \rangle \Rightarrow_{cont} \langle k, i + 1 \rangle$
$\langle stop, v \rangle \Rightarrow_{final} v$

```

where  $e_{base} = e_{empty}[succ \mapsto succ]$ 
 $e_{init} = e_{base}$ 

```

4.7. Summary and conclusion

We have presented a series of evaluators and one abstract machine that correspond to a call-by-value monadic evaluator and the identity monad. The first evaluator is a traditional, Lisp-like one in direct style. The machine is the CEK machine. The correctness of the evaluators and of the abstract machine is a corollary of the correctness of the original monadic evaluator and of the transformations.

5. From a lifted state monad to an abstract machine

We specify a lifted state monad and inline its components in the monadic evaluator, obtaining an evaluator in state-passing style. Closure converting, CPS-transforming, and defunctionalizing this state-passing evaluator yields a version of the CEK machine with error and state. This monad and this machine form a stepping stone towards stack inspection.

5.1. A lifted state monad

We consider a lifted state monad where the state is, for conciseness, one integer. We equip this monad with three operations for reading and writing the state and for failing:

```

signature LIFTED_STATE_MONAD
= sig
  include MONAD
  type storable
  type state

  val get : storable monad
  val set : storable -> storable monad
  val fail : 'a monad
end

structure Lifted_State_Monad : LIFTED_STATE_MONAD
= struct
  datatype 'a lift = LIFT of 'a | BOTTOM
  type storable = int
  type state = storable
  type 'a monad = state -> ('a * state) lift

  fun unit a = (fn s => LIFT (a, s))
  fun bind (m, k) = (fn s => case m s
                             of (LIFT (a, s')) => k a s'
                              | BOTTOM => BOTTOM)

  val get = (fn s => LIFT (s, s))
  fun set i = (fn s => LIFT (s, i))
  val fail = (fn s => BOTTOM)
end

```

Proposition 1. *The type constructor above, together with the above definitions of unit and bind, satisfies the three monadic laws.*

Proof. The lifted state monad is a combination of the lifting monad and of a state monad, and is known to be a monad [28]. Alternatively, the monadic laws can be verified by equational reasoning. \square

We extend the base environment with three functions `get`, `set`, and `fail`:

```
val env_init
  = Env.extend ("fail", FUN (fn _ => fail),
    Env.extend ("set", FUN (fn (NUM i)
      => bind (set i, fn i => unit (NUM i))),
    Env.extend ("get", FUN (fn _ => bind (get, fn i => unit (NUM i))),
    env_base)))
```

Evaluation starts with an initial state `state_init`: `Lifted_State_Monad.state`.

5.2. A CEK machine with error and state

Inlining the components of the lifted state monad in the monadic evaluator of Section 2 and uncurrying the `eval` function and the function space in the data type of expressible values yields a call-by-value evaluator in state-passing style. As in Section 4, we closure-convert, CPS-transform, and defunctionalize this inlined evaluator. The result is a version of the CEK machine with error and state [16]. The source language and evaluation contexts are as in the CEK machine of Section 4.

- Expressible values (integers, closures, and predefined functions) and results:

$$v ::= i \mid [x, t, e] \mid succ \mid get \mid set \mid fail$$

$$r ::= lift(v, s) \mid bottom$$

- Initial transition, transition rules (two kinds), and final transition:

$t \Rightarrow_{init} \langle t, e_{init}, s_{init}, stop \rangle$
$\langle \ulcorner i \urcorner, e, s, k \rangle \Rightarrow_{eval} \langle k, lift(i, s) \rangle$
$\langle x, e, s, k \rangle \Rightarrow_{eval} \langle k, lift(e(x), s) \rangle$
$\langle \lambda x.t, e, s, k \rangle \Rightarrow_{eval} \langle k, lift([x, t, e], s) \rangle$
$\langle t_0 t_1, e, s, k \rangle \Rightarrow_{eval} \langle t_0, e, s, arg(t_1, e, k) \rangle$
$\langle arg(t_1, e, k), lift(v, s) \rangle \Rightarrow_{cont} \langle t_1, e, s, fun(v, k) \rangle$
$\langle arg(t_1, e, k), bottom \rangle \Rightarrow_{cont} \langle k, bottom \rangle$
$\langle fun([x, t, e], k), lift(v, s) \rangle \Rightarrow_{cont} \langle t, e[x \mapsto v], s, k \rangle$
$\langle fun(succ, k), lift(i, s) \rangle \Rightarrow_{cont} \langle k, lift(i + 1, s) \rangle$
$\langle fun(get, k), lift(v, s) \rangle \Rightarrow_{cont} \langle k, lift(s, s) \rangle$
$\langle fun(set, k), lift(i, s) \rangle \Rightarrow_{cont} \langle k, lift(s, i) \rangle$
$\langle fun(fail, k), lift(v, s) \rangle \Rightarrow_{cont} \langle k, bottom \rangle$
$\langle fun(v, k), bottom \rangle \Rightarrow_{cont} \langle k, bottom \rangle$
$\langle stop, r \rangle \Rightarrow_{final} r$

where $e_{base} = e_{empty}[succ \mapsto succ]$
 $e_{init} = e_{base}[get \mapsto get][set \mapsto set][fail \mapsto fail]$

and s_{init} is the initial state.

In case of failure, the machine propagates *bottom* out of the surrounding evaluation contexts and yields it as the final result. The machine could be optimized by re-classifying the *fail*-transition to be a final transition (i.e., a transition that directly yields *bottom* as the result) and by removing all the *bottom*-propagating transitions. In the corresponding CPS evaluator, this optimization hinges on the type isomorphism between the sum-accepting continuation $((\text{value} * \text{state}) \text{ lift}) \rightarrow 'a$ and the pair of continuations $(\text{value} * \text{state} \rightarrow 'a) * (\text{unit} \rightarrow 'a)$. This isomorphism enables the optimization from $\text{unit} \rightarrow 'a$ (i.e., a propagating continuation) to $'a$ (i.e., an immediate stop). We illustrate this optimization in Appendix A.

5.3. Summary and conclusion

We have presented a lifted state monad and the abstract machine that corresponds to the call-by-value monadic evaluator and this monad. The evaluator obtained by inlining the components of the lifted state monad is in state-passing style. The machine is a version of the CEK machine with state and explicit error propagation. The correctness of the evaluators and of the abstract machine is a corollary of the correctness of the original monadic evaluator and of the transformations.

6. Stack inspection as a lifted state monad

Stack inspection is a security mechanism developed to allow code with different levels of trust to interact in the same execution environment (e.g., the JVM or the CLR) [20]. Before execution, the code is annotated with a subset R of a fixed set of permissions P . For example, trusted code is annotated with all permissions and untrusted code is only annotated with a subset of permissions. Before accessing a restricted resource during execution, the call stack is inspected to test that the required access permissions are available. This test consists of traversing the entire call stack to ensure that the direct caller and all indirect callers all have the required permissions to access the resource. Traversing the entire call stack prevents untrusted code from gaining access to restricted resources by (indirectly) calling trusted code. Trusted code can prevent inspection of its callers for some permissions by explicitly granting those permissions. Trusted code can only grant permissions with which it has been annotated.

Because the entire call stack has to be inspected before accessing resources, the stack-inspection mechanism seems to be incompatible with global tail-call optimization. However, Clements and Felleisen [8] have shown that this is not true and that stack inspection is in fact compatible with global tail-call optimization. Their observation is that the security information of multiple tail calls can be summarized in a permission table. If each stack frame contains a permission table, stack frames do not need to be allocated for tail-calls—the permission table of the current stack frame can be updated instead. This tail-recursive semantics for stack inspection is similar to tail-call optimization in (dynamically scoped) Lisp [31]. It is presented in the form of a CESK machine, the CM machine, and Clements and Felleisen have proved that this machine uses asymptotically as much space as Clinger's [9] tail-call optimized CESK machine. In the CM machine, the call stack is represented as CEK evaluation contexts enriched with a permission table.

The language of the CM machine is the λ -calculus extended with four constructs:

- (1) $R[t]$, to annotate a term t with a set of permissions R . When executed, the permissions available are restricted to the permissions in R by making the complement $\bar{R} = P \setminus R$ unavailable; t is then executed with the updated permissions.
- (2) $\text{grant } R \text{ in } t$, to grant a set of permissions R during the evaluation of a term t . When executed, the permissions R are made available, and t is executed with the updated permissions.
- (3) $\text{test } R \text{ then } t_0 \text{ else } t_1$, to branch depending on whether a set of permissions R is available. When executed, the call stack is inspected using a predicate called \mathcal{OK} , and t_0 is executed if the permissions are available; otherwise t_1 is executed.
- (4) fail , to fail due to a security error. When executed, the evaluation is terminated with a security error (and therefore the machine is optimized as described in Appendix A).

Our starting point is a simplified version of Clements and Felleisen's CM machine. Their machine includes a heap and a garbage-collection rule to make it possible to extend Clinger's space-complexity analysis to the CM machine. For simplicity, we leave out the heap and the garbage-collection rule from the machine, and, without loss of generality (because the source language is untyped), we omit recursive functions from the source language. Clements and Felleisen's source language does not have literals; for simplicity, we do likewise and we omit literals and the successor function from the source language. Our focus is the basic stack-inspection mechanism and the features we have omitted from the CM machine do not interfere with this basic mechanism. The simplified CM machine is as follows:

- Permissions $R \subseteq P$ and permission tables $m \in P \rightarrow \{\text{grant}, \text{no}\}$ for a fixed set of permissions P .

- Source syntax (terms):

$$t ::= x \mid \lambda x.t \mid t_0 t_1 \mid R[t] \mid \text{grant } R \text{ in } t \mid \text{test } R \text{ then } t_0 \text{ else } t_1 \mid \text{fail}$$

- Expressible values (closures), outcomes, and evaluation contexts:

$$\begin{aligned} v &::= [x, t, e] \\ o &::= v \mid \text{fail} \\ k &::= \text{stop}(m) \mid \text{arg}(t, e, k, m) \mid \text{fun}(v, k, m) \end{aligned}$$

- Initial transition, transition rules (two kinds), and final transitions:

$t \Rightarrow_{\text{init}} \langle t, e_{\text{empty}}, \text{stop}(m_{\text{empty}}) \rangle$
$\langle x, e, k \rangle \Rightarrow_{\text{eval}} \langle k, e(x) \rangle$
$\langle \lambda x.t, e, k \rangle \Rightarrow_{\text{eval}} \langle k, [x, t, e] \rangle$
$\langle t_0 t_1, e, k \rangle \Rightarrow_{\text{eval}} \langle t_0, e, \text{arg}(t_1, e, k, m_{\text{empty}}) \rangle$
$\langle R[t], e, k \rangle \Rightarrow_{\text{eval}} \langle t, e, k[\bar{R} \mapsto \text{no}] \rangle$
$\langle \text{grant } R \text{ in } t, e, k \rangle \Rightarrow_{\text{eval}} \langle t, e, k[R \mapsto \text{grant}] \rangle$
$\langle \text{test } R \text{ then } t_0 \text{ else } t_1, e, k \rangle \Rightarrow_{\text{eval}} \langle t_0, e, k \rangle \text{ if } \mathcal{OK}[R][k]$
$\langle \text{test } R \text{ then } t_0 \text{ else } t_1, e, k \rangle \Rightarrow_{\text{eval}} \langle t_1, e, k \rangle \text{ if not } \mathcal{OK}[R][k]$
$\langle \text{fail}, e, k \rangle \Rightarrow_{\text{final}} \text{fail}$
$\langle \text{arg}(t, e, k, m), v \rangle \Rightarrow_{\text{cont}} \langle t, e, \text{fun}(v, k, m_{\text{empty}}) \rangle$
$\langle \text{fun}([x, t, e], k, m), v \rangle \Rightarrow_{\text{cont}} \langle t, e[x \mapsto v], k \rangle$
$\langle \text{stop}(m), v \rangle \Rightarrow_{\text{final}} v$

where m_{empty} denotes the empty permission table,

$$\begin{aligned}\text{stop}(m)[R \mapsto c] &= \text{stop}(m[R \mapsto c]) \\ \text{arg}(t, e, k, m)[R \mapsto c] &= \text{arg}(t, e, k, m[R \mapsto c]) \\ \text{fun}(v, k, m)[R \mapsto c] &= \text{fun}(v, k, m[R \mapsto c])\end{aligned}$$

and

$$\begin{aligned}\mathcal{OK}[\emptyset][k] &= \text{true} \\ \mathcal{OK}[R][\text{stop}(m)] &= R \cap m^{-1}(\text{no}) = \emptyset \\ \left. \begin{aligned}\mathcal{OK}[R][\text{arg}(t, e, k, m)] \\ \mathcal{OK}[R][\text{fun}(t, k, m)]\end{aligned} \right\} &= (R \cap m^{-1}(\text{no}) = \emptyset) \wedge \mathcal{OK}[R \setminus m^{-1}(\text{grant})][k]\end{aligned}$$

In the CM machine, evaluation contexts are CEK evaluation contexts enriched with permission tables. They are therefore a zipped version of the CEK evaluation contexts and a stack of permission tables. We unzip the CM evaluation contexts into CEK evaluation contexts and a stack of permission tables. This unzipping corresponds to separating the security mechanism from the function call mechanism. In the literature, it has been argued that security mechanisms such as stack inspection are best viewed separately from the stack. For instance, Abadi and Fournet [1] separate the security mechanism from the stack in order to obtain a stronger security mechanism that is not tied to the behaviour of the stack. Wallach et al. [38] also separate the security mechanism from the stack to obtain an alternative semantics and implementation of stack inspection. As for us, we separate the security mechanism from the stack in order to make the evaluation mechanism clearer: the CM machine is a variant of the CEK machine that keeps track of a stack of permission tables.

The unzipped CM machine is as follows. Permissions, permission tables, source syntax, expressible values, and outcomes remain the same as in the original CM machine. The \mathcal{OK} predicate is changed to inspect the stack of permission tables instead of the evaluation contexts:

- Evaluation contexts:

$$k ::= \text{stop} \mid \text{arg}(t, e, k) \mid \text{fun}(v, k)$$

- Initial transition, transition rules (two kinds), and final transitions:

$t \Rightarrow_{\text{init}} \langle t, e_{\text{empty}}, m_{\text{empty}} :: \text{nil}, \text{stop} \rangle$
$\langle x, e, ms, k \rangle \Rightarrow_{\text{eval}} \langle k, ms, e(x) \rangle$
$\langle \lambda x.t, e, ms, k \rangle \Rightarrow_{\text{eval}} \langle k, ms, [x, t, e] \rangle$
$\langle t_0 t_1, e, ms, k \rangle \Rightarrow_{\text{eval}} \langle t_0, e, m_{\text{empty}} :: ms, \text{arg}(t_1, e, k) \rangle$
$\langle R[t], e, m :: ms, k \rangle \Rightarrow_{\text{eval}} \langle t, e, m[\bar{R} \mapsto \text{no}] :: ms, k \rangle$
$\langle \text{grant } R \text{ in } t, e, m :: ms, k \rangle \Rightarrow_{\text{eval}} \langle t, e, m[R \mapsto \text{grant}] :: ms, k \rangle$
$\langle \text{test } R \text{ then } t_0 \text{ else } t_1, e, ms, k \rangle \Rightarrow_{\text{eval}} \langle t_0, e, ms, k \rangle \text{ if } \mathcal{OK}[R][ms]$
$\langle \text{test } R \text{ then } t_0 \text{ else } t_1, e, ms, k \rangle \Rightarrow_{\text{eval}} \langle t_1, e, ms, k \rangle \text{ if not } \mathcal{OK}[R][ms]$
$\langle \text{fail}, e, ms, k \rangle \Rightarrow_{\text{final}} \text{fail}$
$\langle \text{arg}(t, e, k), m :: ms, v \rangle \Rightarrow_{\text{cont}} \langle t, e, m_{\text{empty}} :: ms, \text{fun}(v, k) \rangle$
$\langle \text{fun}([x, t, e], k), m :: ms, v \rangle \Rightarrow_{\text{cont}} \langle t, e[x \mapsto v], ms, k \rangle$
$\langle \text{stop}, ms, v \rangle \Rightarrow_{\text{final}} v$

where

$$\begin{aligned} \mathcal{OK}[\emptyset][ms] &= \text{true} \\ \mathcal{OK}[R][nil] &= \text{true} \\ \mathcal{OK}[R][m :: ms] &= (R \cap m^{-1}(no) = \emptyset) \wedge \mathcal{OK}[R \setminus m^{-1}(grant)][ms] \end{aligned}$$

As we have already observed in previous work [3,7,10,12,13], the evaluation contexts, together with the *cont* transition function, are the defunctionalized counterpart of a continuation. We can therefore “refunctionalize” this continuation and then write the evaluator in direct style. The resulting evaluator threads a state—the stack of permission tables—and evaluation can fail. The evaluator can therefore be expressed as an instance of a monadic evaluator with a lifted state monad.

In the lifted state monad for stack inspection, the storable values are permission tables, and the state is a stack of storable values. The operations on the permission tables are expressed as the monadic operations `push_empty`, `pop_top`, `clear_top`, `mark_complement_no`, `mark_grant`, and `OK`. Furthermore, the monadic operation `fail` terminates the computation with a security error. The stack-inspection state monad is implemented as a structure with the following signature:

```
signature STACK_INSPECTION_LIFTED_STATE_MONAD
= sig
  include MONAD
  val fail : 'a monad
  val push_empty : unit monad
  val pop_top : unit monad
  val clear_top : unit monad
  val mark_complement_no : permission Set.set -> unit monad
  val mark_grant : permission Set.set -> unit monad
  val OK : permission Set.set -> bool monad
end
```

where `permission` is a type of permissions and `Set.set` is a polymorphic type of sets.

The definitions of `unit` and `bind` are those of the lifted state monad of Section 5: `fail` implements the security error; `push_empty` pushes an empty permission table on top of the permission-table stack; `pop_top` pops the top permission table off the permission-table stack; `clear_top` clears the topmost permission table; `mark_complement_no` updates the topmost permission table by making the complement of the argument set of permissions unavailable; `mark_grant` updates the topmost permission table by making the argument set of permissions available; and `OK` inspects the permission stack to test whether the argument permissions are available.

The source language is represented as an ML datatype:

```
datatype term = VAR of ide
              | LAM of ide * term
              | APP of term * term
              | FRAME of permission Set.set * term
              | GRANT of permission Set.set * term
              | TEST of permission Set.set * term * term
              | FAIL
```

The monadic evaluator corresponding to the unzipped version of the CM machine reads as follows:

```

datatype value = FUN of value -> value monad

(* eval : term * value Env.env -> value monad *)
fun eval (LAM (x, t), e)
  = unit (FUN (fn v => eval (t, Env.extend (x, v, e))))
| eval (VAR x, e)
  = unit (Env.lookup (e, x))
| eval (APP (t0, t1), e)
  = bind (push_empty, fn () =>
    bind (eval (t0, e), fn v0 =>
      bind (clear_top, fn () =>
        bind (eval (t1, e), fn v1 =>
          bind (pop_top, fn () => let val (FUN f) = v0
                                in f v1
                                end))))))
| eval (FRAME (R, t), e)
  = bind (mark_complement_no R, fn () => eval (t, e))
| eval (GRANT (R, t), e)
  = bind (mark_grant R, fn () => eval (t, e))
| eval (TEST (R, t0, t1), e)
  = bind (OK R, fn b => if b then eval (t0, e) else eval (t1, e))
| eval (FAIL, e)
  = fail

```

This evaluator alters the state by pushing and popping permission tables when evaluating applications. One could be tempted to make these changes implicit by integrating them in the definition of `bind` and use the generic evaluator of Section 2. However, the result would not be a monad because the right-unit law would not hold. Therefore, the state changes have to appear explicitly in the monadic evaluator—a situation that has precedents, e.g., in one of Wadler’s monadic evaluators [37, Section 2.5]. For these reasons the evaluator just above differs from the generic evaluator of Section 2.

The derivation process is reversible. Starting from this lifted state monad where the state is a stack of permission tables and this monadic evaluator, it is a simple exercise to reconstruct the unzipped CM machine by inlining the monad, closure converting the expressible values, CPS-transforming the evaluator, optimizing the continuation as illustrated in Appendix A to stop immediately in case of failure, and defunctionalizing the resulting continuations. In addition, we are now in position to combine properly tail-recursive stack inspection with other effects by combining the stack-inspection monad with other monads at the monadic level. Inlining such combined monads lets us derive abstract machines with properly tail-recursive stack inspection and other computational effects [5].

To summarize, we have shown that Clements and Felleisen’s properly tail-recursive stack inspection can be expressed as a lifted state monad. Constructing abstract machines for a language with stack inspection and other effects expressed as monads therefore reduces to designing the desired combination of the monads and then mechanically deriving the corresponding abstract machine. The correctness of this abstract machine is a corollary of the correctness of the original monadic evaluator and of the transformations.

7. A dedicated monad for stack inspection

We observe that the lifted state monad is overly general to characterize the computational behaviour of stack inspection:

```
type 'a monad = permission_table list -> ('a * permission_table list) lift
```

This type would also fit if all permissions in the stack were updatable. However, that is not the case—only the top permission table can be modified, and the other permission tables in the stack are read-only.

Instead, we can cache the top permission table and make it both readable and writable while keeping the rest of the stack read-only. The corresponding type constructor reads as follows:

```
type 'a monad = permission_table * permission_table list
              -> ('a * permission_table) lift
```

Proposition 2. *The type constructor above, together with the following definitions of `unit` and `bind`, satisfies the three monadic laws.*

```
fun unit a = (fn (p, pl) => LIFT (a, p))
fun bind (m, k) = (fn (p, pl) => case m (p, pl)
                                of (LIFT (a, p'))
                                 => k a (p', pl)
                                | BOTTOM
                                 => BOTTOM)
```

Proof. By equational reasoning. \square

This monad provides a more accurate characterization of stack inspection than the one in Section 6.

As an exercise, we have constructed the corresponding abstract machine. This machine is similar to the one in Section 6.

8. Related work

Stack inspection is used as a fine-grained access control mechanism for Java [22]. It allows code with different levels of trust to safely interact in the same execution environment. Before access to a restricted resource is allowed, the entire call stack is inspected to test that the required permissions are available. Wallach et al. [38] present a semantics for stack inspection based on a belief logic. Their semantics is not tied to inspecting stack frames, and it is thus compatible with tail-call optimization. Their implementation, called security-passing style, allows them to implement stack inspection for Java without changing the JVM. Instead, they perform a global byte-code rewriting before loading. Fournet and Gordon [20] develop a formal semantics and an equational theory for a λ -calculus model of stack inspection. They use this equational theory to formally investigate how stack inspection affects known program transformations such as inlining and tail-call optimization. Clements and Felleisen [8] present a properly tail-call optimized semantics for stack inspection based

on Fournet and Gordon’s semantics. This tail-call optimized semantics is given in the form of a CESK machine, which was the starting point for our work.

Since Moggi’s breakthrough [28], monads have been widely used to parameterize functional programs with effects [6,37]. We are not aware, though, of the use of monads in connection with abstract machines for computational effects.

For several decades abstract machines have been an active area of research, ranging from Landin’s classical SECD machine [25,32] to the modern JVM [26]. As observed by Diehl et al. [15], research on abstract machines has chiefly focused on developing new machines and proving them correct. The thrust of our work is to explore a correspondence between interpreters and abstract machines [3,4,7,10] that takes its roots in Reynold’s seminal work on definitional interpreters [33].

There are two forerunners to our work:

- (1) Reynolds’s original work [33], where he CPS-transforms and defunctionalizes a call-by-value evaluator for lambda-terms. We observe that the resulting first-order evaluator coincides with (and anticipates) the CEK machine.
- (2) Schmidt’s PhD work [34], where he constructs a transition system by defunctionalizing a continuation-passing call-by-name evaluator for lambda-terms. We observe that the resulting transition system coincides with (and anticipates) Krivine’s machine.

The present work is a next step of our study of the correspondence between evaluators and abstract machines. Essentially the same correspondence has been put to use by Graunke et al. [23] to transform functional programs into abstract machines for programming the web. The only difference is that Graunke, Findler, Krishnamurthi, and Felleisen use lambda-lifting instead of closure conversion. They do not need closure conversion because they do not consider evaluators for higher-order programming languages, and we do not need lambda-lifting because our evaluators are already lambda-lifted [14,24].

9. Conclusion

We have extended the correspondence between evaluators and abstract machines from the pure setting of the λ -calculus to the impure setting of the computational λ -calculus. Throughout, we have advocated that a viable alternative to designing abstract machines for languages with computational effects on a case-by-case basis is deriving them from a monadic evaluator and a computational monad. As a consequence one does not need to establish the correctness of each abstract machine on a case-by-case basis since it is a corollary of the correctness of the original monadic evaluator and of the transformations. We have illustrated this alternative with several monads.

We have also characterized Clements and Felleisen’s properly tail-recursive stack inspection as a lifted state monad, and we have proposed an alternative, dedicated monad for this computational effect. These two monads enable us to combine stack inspection with other computational effects at the monadic level and then systematically construct the corresponding abstract machines. We are therefore in position to construct, e.g., a variant of Krivine’s machine with stack inspection as well as variants of the Categorical Abstract Machine and of the SECD machine with arbitrary computational effects expressed as monads.

Acknowledgements

We are grateful to Dariusz Biernacki, Julia Lawall, Peter Thiemann, and Mitchell Wand for commenting a preliminary version of this article. Thanks are also due to Eugenio Moggi for his editorship and to John Clements and the anonymous referees for their feedback.

This work is partially supported by the ESPRIT Working Group APPSEM II (<http://www.appsem.org>), the SECURE project EU FET-GC IST-2001-32486, and the Danish Natural Science Research Council, Grant No. 21-03-0545.

An extended version of this article is available in the BRICS Research Series [5].

Appendix A. Propagating vs. stopping

This appendix illustrates the optimization of returning a final result directly instead of propagating it through surrounding evaluation contexts. We consider the traditional example of multiplying the leaves of a tree of integers:

```
datatype bt = LEAF of int
           | NODE of bt * bt
```

We want to take advantage of the fact that 0 is an absorbant element for multiplication. To this end, we lift the intermediate results of the auxiliary function that traverses the input tree:

```
datatype int_lifted = ZERO
                  | NOT_ZERO of int
```

```
(* mult_ds : bt -> int *)
fun mult_ds t
  = let (* visit : bt -> int_lifted *)
      fun visit (LEAF 0) = ZERO
        | visit (LEAF n) = NOT_ZERO n
        | visit (NODE (t1, t2))
          = (case visit t1
              of ZERO
               => ZERO
              | (NOT_ZERO n1)
               => (case visit t2
                    of ZERO
                     => ZERO
                    | (NOT_ZERO n2)
                     => NOT_ZERO (n1 * n2)))
          in case visit t
              of ZERO => 0
               | (NOT_ZERO n) => n
          end
```

If a 0 leaf is encountered during the recursive descent, ZERO is propagated out until the top-level case expression.

Let us write `visit` in CPS:

```
(* mult_cps : bt -> int *)
fun mult_cps t
  = let (* visit : bt * (int_lifted -> int) -> int *)
      fun visit (LEAF 0, k) = k ZERO
        | visit (LEAF n, k) = k (NOT_ZERO n)
        | visit (NODE (t1, t2), k)
          = visit (t1, fn ZERO
                  => k ZERO
                  | (NOT_ZERO n1)
                  => visit (t2, fn ZERO
                            => k ZERO
                            | (NOT_ZERO n2)
                            => k (NOT_ZERO (n1 * n2))))
      in visit (t, fn ZERO => 0
                | (NOT_ZERO n) => n)
      end
```

The same propagation takes place. To optimize it, we use the type isomorphism between the sum-accepting continuation `int_lifted -> int` and the pair of continuations `(unit -> int) * (int -> int)`, one for propagating the final result and one to continue the computation, and we simplify the propagating continuation away:

```
(* mult_cps_opt : bt -> int *)
fun mult_cps_opt t
  = let (* visit : bt * (int -> int) -> int *)
      fun visit (LEAF 0, k) = 0
        | visit (LEAF n, k) = k n
        | visit (NODE (t1, t2), k)
          = visit (t1, fn n1 => visit (t2, fn n2 => k (n1 * n2)))
      in visit (t, fn n => n)
      end
```

In the optimized version, the continuation is only applied to non-zero intermediate results, and as soon as a 0 leaf is encountered, the computation stops.

References

- [1] M. Abadi, C. Fournet, Access control based on execution history, in: V. Gligor, M. Reiter (Eds.), Proc. of the 10th Annu. Network and Distributed System Security Symp. (NDSS'03), San Diego, California, Internet Society, February 2003, pp. 107–121.
- [2] M.S. Ager, D. Biernacki, O. Danvy, J. Midtgaard, From interpreter to compiler and virtual machine: a functional derivation, Technical Report BRICS RS-03-14, DAIMI, Department of Computer Science, University of Aarhus, Aarhus, Denmark, March 2003.
- [3] M.S. Ager, D. Biernacki, O. Danvy, J. Midtgaard, A functional correspondence between evaluators and abstract machines, in: D. Miller (Ed.), Proc. of the Fifth ACM-SIGPLAN Internat. Conf. on Principles and Practice of Declarative Programming (PPDP'03), ACM Press, New York, August 2003, pp. 8–19.
- [4] M.S. Ager, O. Danvy, J. Midtgaard, A functional correspondence between call-by-need evaluators and lazy abstract machines, Inform. Process. Lett. 90 (5) (2004) 223–232 (Extended version available as the technical report BRICS-RS-04-3).

- [5] M.S. Ager, O. Danvy, J. Midtgaard, A functional correspondence between monadic evaluators and abstract machines for languages with computational effects (extended version), Technical Report BRICS RS-04-28, DAIMI, Department of Computer Science, University of Aarhus, Aarhus, Denmark, December 2004.
- [6] N. Benton, J. Hughes, E. Moggi, Monads and effects, in: G. Barthe, P. Dybjer, L. Pinto, J. Saraiva (Eds.), Applied Semantics—Advanced Lectures, Vol. 2395, Lecture Notes in Computer Science, Caminha, Portugal, Springer, Berlin, September 2000, pp. 42–122.
- [7] D. Biernacki, O. Danvy, From interpreter to logic engine by defunctionalization, in: M. Bruynooghe (Ed.), Logic Based Program Synthesis and Transformation, 13th Internat. Symp., LOPSTR 2003, Vol. 3018, Lecture Notes in Computer Science, Uppsala, Sweden, Springer, Berlin, August 2003, pp. 143–159.
- [8] J. Clements, M. Felleisen, A tail-recursive semantics for stack inspections, in: P. Degano (Ed.), Programming Languages and Systems, 12th European Symp. on Programming, ESOP 2003, Vol. 2618, Lecture Notes in Computer Science, Warsaw, Poland, Springer, Berlin, April 2003, pp. 22–37.
- [9] W.D. Clinger, Proper tail recursion and space efficiency, in: K.D. Cooper (Ed.), Proc. of the ACM SIGPLAN'98 Conf. on Programming Languages Design and Implementation, Montréal, Canada, ACM Press, New York, June 1998, pp. 174–185.
- [10] O. Danvy, A rational deconstruction of Landin's SECD machine, in: C. Grelck, F. Huch, G.J. Michaelson, P. Trinder (Eds.), Implementation and Application of Functional Languages, 16th Internat. Workshop, IFL'04, Vol. 3474, Lecture Notes in Computer Science, Lübeck, Germany, Springer, Berlin, September 2004, pp. 52–71 (Extended version available as the technical report BRICS-RS-03-33).
- [11] O. Danvy, A. Filinski, Representing control, a study of the CPS transformation, *Math. Struct. Comput. Sci.* 2 (4) (1992) 361–391.
- [12] O. Danvy, L.R. Nielsen, Defunctionalization at work, in: H. Søndergaard (Ed.), Proc. of the Third Internat. ACM SIGPLAN Conf. on Principles and Practice of Declarative Programming (PPDP'01), Firenze, Italy, ACM Press, New York, September 2001, pp. 162–174, (Extended version available as the technical report BRICS RS-01-23).
- [13] O. Danvy, L.R. Nielsen, Refocusing in reduction semantics, Technical Report BRICS RS-04-26, DAIMI, Department of Computer Science, University of Aarhus, Aarhus, Denmark, November 2004. A preliminary version appears in the informal Proc. of the Second Internat. Workshop on Rule-Based Programming (RULE 2001), Electronic Notes in Theoretical Computer Science, Vol. 59.4.
- [14] O. Danvy, U.P. Schultz, Lambda-lifting in quadratic time, *J. Funct. Logic Programming* 10 (1) (2004) (Available online at <http://danae.uni-muenster.de/lehre/kuchen/JFLP/>).
- [15] S. Diehl, P. Hartel, P. Sestoft, Abstract machines for programming language implementation, *Future Generation Comput. Systems* 16 (2000) 739–751.
- [16] M. Felleisen, M. Flatt, Programming languages and lambda calculi. Unpublished lecture notes. <http://www.ccs.neu.edu/home/matthias/3810-w02/readings.html>, 1989–2003.
- [17] M. Felleisen, D.P. Friedman, Control operators, the SECD machine, and the λ -calculus, in: M. Wirsing (Ed.), Formal Description of Programming Concepts III, Elsevier Science Publishers B.V., North-Holland, Amsterdam, 1986, pp. 193–217.
- [18] A. Filinski, Representing monads, in: H.-J. Boehm (Ed.), Proc. of the Twenty-First Annu. ACM Symp. on Principles of Programming Languages, Portland, Oregon, ACM Press, New York, January 1994, pp. 446–457.
- [19] C. Flanagan, A. Sabry, B.F. Duba, M. Felleisen, The essence of compiling with continuations, in: D.W. Wall (Ed.), Proc. of the ACM SIGPLAN'93 Conf. on Programming Languages Design and Implementation, SIGPLAN Notices, Vol. 28(6), Albuquerque, New Mexico, ACM Press, New York, June 1993, pp. 237–247.
- [20] C. Fournet, A.D. Gordon, Stack inspection: theory and variants, *ACM Trans. Programming Languages Systems* 25 (3) (2003) 360–399.
- [21] D.P. Friedman, M. Wand, C.T. Haynes, Essentials of Programming Languages, second ed., MIT Press, Cambridge, MA, 2001.
- [22] L. Gong, R. Schemers, Implementing protection domains in Java Development Kit 1.2, in: Proc. of the Internat. Symp. on Network and Distributed System Security, San Diego, California, March 1998.
- [23] P.T. Graunke, R.B. Findler, S. Krishnamurthi, M. Felleisen, Automatically restructuring programs for the web, in: M.S. Feather, M. Goedicke (Eds.), 16th IEEE Internat. Conf. on Automated Software Engineering (ASE 2001), Coronado Island, San Diego, California, USA, IEEE Computer Society, Silver Spring, MD, November 2001, pp. 211–222.

- [24] T. Johnsson, Lambda lifting: transforming programs to recursive equations, in: J.-P. Jouannaud (Ed.), *Functional Programming Languages and Computer Architecture*, Vol. 201, Lecture Notes in Computer Science, Nancy, France, Springer, Berlin, September 1985, pp. 190–203.
- [25] P.J. Landin, The mechanical evaluation of expressions, *The Comput. J.* 6 (4) (1964) 308–320.
- [26] T. Lindholm, F. Yellin, *The Java™ Virtual Machine Specification*, second ed., Addison-Wesley, Reading, MA, 1999.
- [27] R. Milner, M. Tofte, R. Harper, D. MacQueen, *The Definition of Standard ML (Revised)*, MIT Press, Cambridge, MA, 1997.
- [28] E. Moggi, Notions of computation and monads, *Inform. and Comput.* 93 (1991) 55–92.
- [29] G.D. Plotkin, Call-by-name, call-by-value and the λ -calculus, *Theoret. Comput. Sci.* 1 (1975) 125–159.
- [30] G.D. Plotkin, A structural approach to operational semantics, Technical Report FN-19, DAIMI, Department of Computer Science, University of Aarhus, Aarhus, Denmark, September 1981.
- [31] C. Queinsec, *Lisp in Small Pieces*, Cambridge University Press, Cambridge, 1996.
- [32] J.D. Ramsdell, The tail-recursive SECD machine, *J. Automated Reasoning* 23 (1) (1999) 43–62.
- [33] J.C. Reynolds, Definitional interpreters for higher-order programming languages, *Higher-Order and Symbolic Comput.* 11 (4) (1998) 363–397 Reprinted from the Proc. of the 25th ACM National Conf. 1972, with a foreword.
- [34] D.A. Schmidt, State transition machines for lambda calculus expressions, in: N.D. Jones (Ed.), *Semantics-Directed Compiler Generation*, Vol. 94, Lecture Notes in Computer Science, Aarhus, Denmark, Springer, Berlin, 1980, pp. 415–440.
- [35] G.L. Steele Jr., *Rabbit: a compiler for scheme*, Master's thesis, Artificial Intelligence Laboratory, Massachusetts Institute of Technology, Cambridge, Massachusetts, May 1978. Technical report AI-TR-474.
- [36] C. Strachey, Fundamental concepts in programming languages, *Higher-Order and Symbolic Comput.* 13 (1/2) (2000) 1–49.
- [37] P. Wadler, The essence of functional programming (invited talk), in: A.W. Appel (Ed.), *Proc. of the Nineteenth Annu. ACM Symp. on Principles of Programming Languages*, Albuquerque, New Mexico, ACM Press, New York, January 1992, pp. 1–14.
- [38] D.S. Wallach, A.W. Appel, E.W. Felten, SAFKASI: a security mechanism for language-based systems, *ACM Trans. Software Eng. Methodol.* 9 (4) (2000) 341–378.